Charge Pump Circuit Optimisation

Charge Pump circuits or “switched capacitor DC converters are a common circuit used in semiconductors to provide different voltage ranges, perhaps where sensors are interfaced to or for low power / sleep modes of operation

Performance measures for switched capacitor circuits might be the power efficiency, duty cycle of the switches and the area of the circuit

Variables of the circuit could be the clock frequency, transistor size and capacitor size

MOTIVATEDs multi objective automated optimisation can easily be used to determine a range of optimal solutions from which the engineer can select the optimum solution

Switched Capacitor – DC converter

Example circuit and netlist with parameterised components for use in MOTIVATED




```
.subckt sw a b ck vss wsw=1u lsw=lmin
ms a ck b vss nch w=wsw l=lsw
.ends

.subckt sc a b c ck1 ck2 vss wsw=1u c=1

xs21 a n1 ck2 vss sw wsw=wsw
xs11 n1 b ck1 vss sw wsw=wsw
xs22 b n2 ck2 vss sw wsw=wsw
xs12 n2 c ck1 vss sw wsw=wsw

xc n1 n2 scap c=c
.ends

.subckt scap a b c=1
cc a b 'c*uc'
.ends

.subckt clock ck vss f=1e6 duty=0.5 trise=1f tfall=1f v0=0.0 v1=1.8
.param period=1.0/f
.param th=duty*period
.param tl=period-th
.param pw=th-trise-tfall

vck ck vss dc=0 pulse('v0' 'v1' 0.0n trise tfall pw period)
.ends
```

Switched Capacitor Test Circuit




```
.param sup=2.0 lmin=100n uc=1p
.param iload=1e-6 cload=10n

.inc genes_params.sp
*.param cfreq=1e6 sw_size=8u cap_size=200

.inc subs.sp

vd vdd 0 dc=sup
xck2 ck2 0 clock v0=sup v1=0.0 f=cfreq
xck1 ck1 0 clock v0=0.0 v1=sup f=cfreq

xsc0 vdd out 0 ck1 ck2 0 sc c=cap_size wsw=sw_size
*xsc1 out out1 0 ck1 ck2 vdd sc c=100

il 0 out iload
cl out 0 cload

.ic v(out)=0
.tran 100n '200.0/cfreq' 0 10n

.control
  run
  source pwM.sp
.endc
```

A Test harness to drive the SC circuit and “measure” the output which MOTIVATED uses as objectives

MOTIVATED can accept hierarchical designs so multiple test benches, and circuits, can be used in the optimisation process

```
* &GENE: 0, 1.0e6, 0.1e6, 10e6,  0.5e6  $ cfreq
* &GENE: 2, 1e-6,  1e-6,  20e-6, 0.5e-6 $ sw_size
* &GENE: 3, 200,   50,    500,   10     $ cap_size

.param cfreq = &0
.param sw_size = &1
.param cap_size = &2
```

“Genes” / parameter file

Setting MOTIVATEDs Inputs/Objectives through the GUI



- Inputs shows the files which have the models or information and those that are to be simulated
- Objectives show what MOTIVATED will optimise and whether to maximise, minimise or have constrained



MOTIVATED GUI – Optimisation process results

- As the Optimiser progresses, the results of each “objective” are put in a table and marked against a reference (initial / typical performance) – green highlights an improvement
- As the number of generations increases, the optimiser selects the “fittest” out of the population to move on to the next generation
- Once all generations are complete, the table displays the resultant best objectives of the final generation

Results Table

Individual	Rank	outrms_obj↑	outpp_obj↔	pe_obj↑	area_obj↓
2516	1	831.222E-3	19.29797E-3	340.8534E-3	400E6
2534	1	829.17E-3	21.03157E-3	290.3991E-3	270E6
2537	1	855.85E-3	17.69018E-3	19.26393E-3	310E6
2547	1	821.611E-3	20.275E-3	470.0782E-3	490E6
2565	1	815.007E-3	22.23073E-3	491.0182E-3	390E6
2565	1	815.007E-3	22.23073E-3	491.0182E-3	390E6
2565	1	815.007E-3	22.23073E-3	491.0182E-3	390E6
2571	1	832.791E-3	46.78597E-3	17.77133E-3	80E6
2573	1	809.696E-3	27.40379E-3	376.1182E-3	180E6
2573	1	809.696E-3	27.40379E-3	376.1182E-3	180E6
2573	1	809.696E-3	27.40379E-3	376.1182E-3	180E6
2580	1	820.206E-3	21.63428E-3	432.009E-3	320E6
2586	1	856.123E-3	17.25609E-3	48.09308E-3	370E6

MOTIVATED GUI – Optimisation process results

- The results can also be displayed during, and at the end of the optimisation as a plot of one objective against another, effectively a pareto curve.
- Each point on the curve represents a fully optimised solution where each objective is at it's maximum and can only be improved at the detriment of a different objective.
- This allows the engineer to select the best solution and trade off of all objectives simply
- In this plot of circuit area vs power efficacy, the optimum would be a solution toward the bottom right